Lab 2. Managing packages and Services



In this lab, you'll learn how to install packages, and learn to control services.

Managing packages

Another key resource type in Puppet is the **package**. A major part of configuring servers by hand involves installing packages, so we will also be using packages a lot in Puppet manifests. Although every operating system has its own package format, and different formats vary quite a lot in their capabilities, Puppet represents all these possibilities with a single package type. If you specify in your Puppet manifest that a given package should be installed, Puppet will use the appropriate package manager commands to install it on whatever platform it's running on.

As you've seen, all resource declarations in Puppet follow this form:

```
RESOURCE_TYPE { TITLE:
   ATTRIBUTE => VALUE,
   ...
}
```

package resources are no different. The RESOURCE_TYPE is package, and the only attribute you usually need to specify is ensure, and the only value it usually needs to take is installed:

```
package { 'cowsay':
   ensure => installed,
}
```

Try this example:

```
puppet apply /examples/package.pp
Notice: Compiled catalog for ubuntu-xenial in environment production in 0.52 seconds
Notice: /Stage[main]/Main/Package[cowsay]/ensure: created
Notice: Applied catalog in 29.53 seconds
```

Let's see whether cowsay is installed:

Now that's a useful package!

How Puppet applies the manifest

The title of the package resource is cowsay, so Puppet knows that we're talking about a package named cowsay.

The ensure attribute governs the installation state of packages: unsurprisingly, installed tells Puppet that the package should be installed.

Exercise

Create a manifest that uses the <code>package</code> resource to install any software you find useful for managing servers. Here are some suggestions: <code>tmux</code> , <code>sysdig</code> , <code>atop</code> , <code>htop</code> , and <code>dstat</code> .

Querying resources with the puppet resource

If you want to see what version of a package Puppet thinks you have installed, you can use the puppet resource tool:

```
puppet resource package openssl
package { 'openssl':
   ensure => '1.1.1f-lubuntu2.2',
}
```

puppet resource TYPE TITLE will output a Puppet manifest representing the current state of the named resource on the system. If you leave out <code>TITLE</code>, you'll get a manifest for all the resources of the type <code>TYPE</code>. For example, if you run <code>puppet resource package</code>, you'll see the Puppet code for all the packages installed on the system.

Note

puppet resource even has an interactive configuration feature. To use it, run the following command:

```
puppet resource -e package openssl
```

If you run this, Puppet will generate a manifest for the current state of the resource, and open it in an editor. If you now make changes and save it, Puppet will apply that manifest to make changes to the system. This is a fun little feature, but it would be rather time-consuming to do your entire configuration this way.

Services

The third most important Puppet resource type is the **service**: a long-running process that either does some continuous kind of work, or waits for requests and then acts on them. For example, on most systems, the process runs all the time and listens for SSH login attempts.

Puppet models services with the service resource type. The service resources look like the following example (you can find this in service.pp in the /examples/ directory. From now on, I'll just give the filename of each example, as they are all in the same directory):

```
service { 'ssh':
  ensure => running,
  enable => true,
}
```

The <code>ensure</code> parameter governs whether the service should be running or not. If its value is <code>running</code>, then as you might expect, Puppet will start the service if it is not running. If you set <code>ensure</code> to <code>stopped</code>, Puppet will stop the service if it is running.

Getting help on resources with puppet describe

If you're struggling to remember all the different attributes of all the different resources, Puppet has a built-in help feature that will remind you. Run the following command, for example:

```
puppet describe service
```

This will give a description of the service resource, along with a complete list of attributes and allowed values. This works for all built-in resource types as well as many provided by third-party modules. To see a list of all the available resource types, run the following command:

```
puppet describe --list
```

The package-file-service pattern

It's very common for a given piece of software to require these three Puppet resource types: the <code>package</code> resource installs the software, the <code>file</code> resource deploys one or more configuration files required for the software, and the <code>service</code> resource runs the software itself.

Here's an example using the MySQL database server (package file service.pp):

```
package { 'mysql-server':
   ensure => installed,
   notify => Service['mysql'],
}

file { '/etc/mysql/mysql.cnf':
   source => '/examples/files/mysql.cnf',
   notify => Service['mysql'],
}

service { 'mysql':
   ensure => running,
   enable => true,
}
```

The package resource makes sure the <code>mysql-server</code> package is installed.

The config file for MySQL is <code>/etc/mysql/mysql.cnf</code>, and we use a <code>file</code> resource to copy this file from the Puppet repo so that we can control MySQL settings.

Finally, the service resource ensures that the <code>mysql</code> service is running.

```
puppet apply /examples/package_file_service.pp
```

Run following command in the terminal to verify that mysql is installed and running:

```
service mysql status
```

Notifying a linked resource

You might have noticed a new attribute, called notify, in the file resource in the previous example:

```
file { '/etc/mysql/mysql.cnf':
   source => '/examples/files/mysql.cnf',
```

```
notify => Service['mysql'],
}
```

What does this do? Imagine you've made a change to the <code>mysql.cnf</code> file and applied this change with Puppet. The updated file will be written to a disk, but because the <code>mysql</code> service is already running, it has no way of knowing that its config file has changed. Therefore, your changes will not actually take effect until the service is restarted. However, Puppet can do this for you if you specify the <code>notify</code> attribute on the <code>file</code> resource. The value of <code>notify</code> is the resource to notify about the change, and what that involves depends on the type of resource that's being notified. When it's a service, the default action is to restart the service.

Resource ordering with require

In the package-file-service example, we declared three resources: the <code>mysql-server</code> package, the <code>/etc/mysql/mysql.cnf</code> file, and the <code>mysql</code> service. If you think about it, they need to be applied in that order. Without the <code>mysql-server</code> package installed, there will be no <code>/etc/mysql/</code> directory to put the <code>mysql.cnf</code> file in. Without the package or the config file, the <code>mysql</code> service won't be able to run.

A perfectly reasonable question to ask is, "Does Puppet apply resources in the same order in which they're declared in the manifest?" The answer is usually yes, unless you explicitly specify a different order, using the require attribute.

All resources support the require attribute, and its value is the name of another resource declared somewhere in the manifest, specified in the same way as when using notify. Here's the package-file-service example again, this time with the resource ordering specified explicitly using require (package file service require.pp):

```
package { 'mysql-server':
    ensure => installed,
}

file { '/etc/mysql/mysql.cnf':
    source => '/examples/files/mysql.cnf',
    notify => Service['mysql'],
    require => Package['mysql-server'],
}

service { 'mysql':
    ensure => running,
    enable => true,
    require => [Package['mysql-server'], File['/etc/mysql/mysql.cnf']],
}
```

You can see that the <code>mysql.cnf</code> resource requires the <code>mysql-server</code> package. The <code>mysql</code> service requires both the other resources, listed as an array within square brackets.

When resources are already in the right order, you don't need to use require, as Puppet will apply the resources in the order you declare them. However, it can be useful to specify an ordering explicitly, for the benefit of those reading the code, especially when there are lots of resources in a manifest file.

Summary

In this lab, we've looked at the common package-file-service pattern and seen how to use the <code>notify</code> attribute on a resource to send a message to another resource indicating that its configuration has been updated. We've covered

the use of the require attribute to make dependencies between resources explicit, when necessary.

You've also learned to use <code>puppet resource</code> to inspect the current state of the system according to Puppet, and <code>puppet describe</code> to get command-line help on all Puppet resources. To check what Puppet would change on the system without actually changing it, we've introduced the <code>--noop</code> and <code>--show_diff</code> options to <code>puppet</code> <code>apply</code>.

In the next lab, we'll see how to use the version control tool Git to keep track of your manifests, we'll get an introduction to fundamental Git concepts, such as the repo and the commit, and you'll learn how to distribute your code to each of the servers you're going to manage with Puppet.